

# Parameterized Models for On-line and Off-line Use

Pieter Wuille and Tom Schrijvers\*

Department of Computer Science, K.U.Leuven, Belgium  
*FirstName.LastName@cs.kuleuven.be*

**Abstract.** The Monadic Constraint Programming framework leverages Haskell’s rich static type system and powerful abstraction mechanisms to implement an embedded domain specific language (EDSL) for constraint programming.

In this paper we show how the same constraint model expressed in the EDSL can be processed in various modes by external constraint solvers. We distinguish between *on-line* and *off-line* use of solvers. In off-line mode, the model is not solved; instead it is compiled to lower-level code that will search for solutions when compiled and run. For on-line use, the search can be handled by either the framework or in the external solver. Off-line mode requires recompilation after each change to the model. To avoid repeated recompilation, we separate model from data by means of parameters that need not be known at compile time. Parametrization poses several challenges, which we resolve by embedding the EDSL more deeply.

## 1 Introduction

The Monadic Constraint Programming framework integrates constraint programming in the functional programming language Haskell [9] as a deeply embedded domain specific language (EDSL). This has a considerable advantage compared to special-purpose Functional Constraint (Logic) Programming (FCP) languages such as Curry [7] or TOY [5]. We directly obtain state-of-the-art functional programming support with zero effort, allowing us to focus on constraint programming itself.

While the integration is not as tight, Haskell does offer good EDSL support to make the embedding quite convenient. Moreover, being less tight does provide for greater flexibility. Aspects that are baked into some FCP languages, such as search strategies or the particular solver used, are much more easily interchanged from within the program. In addition, the deep embedding of the EDSL allows us to use the constraint model for more than straight (on-line) solving. For instance, transformations can be applied to the model for optimization purposes or to better target a particular constraint solver. Alternatively, the model does not have to be solved on-line, but can drive a code generator that produces an executable for off-line solving.

---

\* Post-Doctoral Researcher of the Research Foundation–Flanders (FWO-Vlaanderen).

This paper reports on the FD-MCP module of the framework, specific to finite domain (FD) solvers. We show how the framework supports different modes of processing an FD model, by both on-line and off-line solvers. Then we identify the need for parametrized models to make the off-line solver approach both more useful and more efficient. We show how the framework is adjusted to support parametrized models, including deeply embedded iteration constructs and indexed collections of constraint variables.

## 2 Monadic Constraint Programming

The MCP [10] framework is a highly generic constraint programming framework for Haskell. It provides abstractions for writing constraint models, constraint solvers and search strategies. This paper focuses on the solving and modeling parts.

### 2.1 Generic Constraint Programming Infrastructure

MCP defines type classes, Haskell’s form of interfaces, for `Solvers` and `Terms`:

```
class Monad s => Solver s where
  type Constraint s :: *
  type Label s :: *
  add :: Constraint s -> s Bool
  run :: s a -> a
  mark :: s (Label s)
  goto :: Label s -> s ()

class Solver s => Term s t where
  newvar :: s t
```

A type that implements the `Solver` type class must provide a type<sup>1</sup> to represent its constraints and labels, an `add` function for adding constraints, a `run` function to extract the results, a `mark` function to create a label of its current state, and a `goto` function to return to a previous state.

A solver type `s` must also be a monad [11]. A monadic value `s a` is an abstraction of a form of computation `s` that yields a result `a`. Constraint solvers are typically computations that thread an implicit state: the constraint store.

A solver also provides one or more types of terms: `Term s t` expresses that `t` is a term type of solver type `s`. Each term type provides a method `newvar` to generate new constraint variables of that type.

MCP also defines a data type `Model`, representing a model tree:

```
data Model s a
  = Return a -- return a value
  | Add (Constraint s) (Model s a) -- add a constraint
```

<sup>1</sup> Implemented using associated types in Haskell

```

| Try (Model s a) (Model s a)      -- disjunction
...

```

The model tree is parametrized in the constraint solver `s` and returned result type `a`. This provides a type-safe way for representing constraint problem models for arbitrary solvers and result types.

On top of the model data type, MCP provides syntactic sugar (functions that construct model trees), such as `exists` (create a variable), `exist n` (create a list of `n` variables), `addC` (add a constraint), `/\` (conjunction), `\/` (disjunction), `conj` (conjunction of list of models), ... Finally, `Model s` is also a monad.

## 2.2 The FD-MCP Module

The FD-MCP framework introduces an extra layer of abstraction between the more generic `Solver` interface of the MCP framework and the concrete solver implementations.

In contrast to MCP's generic `Solver` interface, which is parametric in the constraint domain, the `FDSolver` interface of FD-MCP is fully aware of the finite domain (FD) constraint domain: both its syntax (terms and constraints) and meaning (constraint theory). It does however make abstraction of the particular FD solver and e.g., propagation techniques used. Hence, it provides a uniform modeling language that abstracts from the syntactic differences between different FD solvers.

On the one hand, this allows the development of solver-independent models, model transformations (e.g., for optimization) and model abstractions (capturing frequently used patterns). On the other hand, specific solvers may focus on the efficient processing of their constraint primitives without worrying about modeling infrastructure.

**FD-MCP Modeling Primitives** The FD-MCP modeling language is built as a wrapper on top of the MCP solver interface. This way, the domain-independent combinators of the MCP framework, such as conjunction (`/\`) and existential quantification `exist` are available for FD models. The FD-MCP modeling language adds FD-specific constructs to that. Advanced FD constructs are defined in terms of a small set of core primitives, resulting in a layered structure.

The core constraints and terms are defined by the `FDConstraint` and `FDEExpr` types respectively:

```

data FDConstraint s
= Less (FDEExpr s) (FDEExpr s)  -- [[Less x y]] ≡ [[x]] < [[y]]
| Diff (FDEExpr s) (FDEExpr s)  -- [[Diff x y]] ≡ [[x]] ≠ [[y]]
| Same (FDEExpr s) (FDEExpr s)  -- [[Same x y]] ≡ [[x]] = [[y]]
| Dom (FDEExpr s) Int Int       -- [[Dom x y z]] ≡ [[x]] ∈ {y, ..., z}
| AllDiff [FDEExpr s]           -- [[AllDiff [x1, ..., xn]] ≡ ∧i≠j xi ≠ xj
| ...

```

```

data FDEExpr s
= Var (FDTerm s)          -- [[Var v]]      ≡ [[v]]
| Const Int               -- [[Const n]]  ≡ n
| Plus (FDEExpr s) (FDEExpr s) -- [[Plus x y]] ≡ [[x]] + [[y]]
| Minus (FDEExpr s) (FDEExpr s) -- [[Minus x y]] ≡ [[x]] - [[y]]
| Mult (FDEExpr s) (FDEExpr s) -- [[Mult x y]] ≡ [[x]] * [[y]]
| ...

```

where the comment after each constructor shows its denotation, and `FDTerm s` refers to the type of the terms used to represent FD variables.

On top of the core primitives, a number of convenient abstractions and syntactic sugar exist. Firstly, standard arithmetic operators and integer literals can be used for `FDEExpr s` thanks to an implementation of Haskell's `Num` type class. Thus `Plus x (Mult (Const 2) y)` can be written succinctly as `x + 2 * y`. More syntactic sugar exists for writing constraints:

```

x @< y      = Add (Less x y) true
x @> y      = y @< x
x @>= y     = x + 1 @> y
x @: (l,u)  = Dom x l u
xs 'allin' d = conj [ x @: d | x <- xs ]
...

```

**Mapping to the solver backend** The backend takes care of compiling an FD-MCP model to a particular FD solver. To enable this compilation, the solver must implement the following `FDSolver` type class (in addition to implementing the `Solver` type class of the MCP framework):

```

class Term s (FDTerm s) => FDSolver s where
  type FDTerm s :: *
  compile_constraint :: FDConstraint s -> Model s Bool

```

The `FDSolver` type class makes two demands of a solver `s`:

- It must provide an (associated) type `FDTerm s` for its terms.
- The function `compile_constraint` must take care of converting from an individual FD-MCP constraint to a model for the solver. Note that, to allow mapping a single FD-MCP constraint to a conjunction of solver constraints involving auxiliary variables, this function returns a model rather than a single constraint. This model is not allowed to contain any disjunctions.

The following law specifies the `compile_constraint` function:

**Definition 1 (Denotation Preservation).** *The `compile_constraint` function preserves denotation iff*

$$[[\cdot]] \circ \text{compile\_constraint} \equiv [[\cdot]]$$

where `[[·]]` maps a model or constraint onto its denotation, i.e., its logical meaning, and `≡` denotes extensional function equality. Two denotations are equal iff they are logically equivalent.

**Integration with MCP** The `FDSolver` type class allows us to define a generic solver `FDWrapper s` that encapsulates the mapping from the generic model to the solver-specific model.

The `FDWrapper s` is an MCP Solver which uses `FDConstraints` as constraints and `FDExprs` as terms.<sup>2</sup>

```

newtype FDWrapper s a = FDWrapper { unwrapFD :: s a }

instance FDSolver s => Solver (FDWrapper s) where
  type Constraint (FDWrapper s) = FDConstraint s
  add c = FDWrapper $ untree $ compile_constraint c
  ...
  run = run . unwrapFD

instance Term (FDWrapper s) (FDTerm t) where
  newvar = FDWrapper $ newvar >>= \x -> return $ Var x
  ...

untree :: Solver s => Model s a -> s a
untree ... = ...

```

The `add` function first converts one `FDConstraint s` to a model tree for the underlying solver `s` with the `compile_constraint` function. Then, it turns this tree into a single (wrapped) monad action for the underlying solver `s` using `untree`. While `untree` is a generic function that works on any `Model`, instances of `FDSolver` provide their own `compile_constraint` to do the translation to their internal constraints. A similar approach is used for the other solver methods. The `newvar` method of the wrapper requests a new variable from the underlying solver, and returns it wrapped in a `Var` constructor. This is why the `FDConstraint` and `FDExpr` structures, as well as `FDWrapper` itself, are parametrized in the underlying MCP solver `s`. Finally, `run` unwraps the encapsulated monad action and runs it.

### 3 Solver Backends and Modes

The initial release of the MCP framework featured only one solver, a simple FD solver implemented in Haskell. However, rather than implement a solver in Haskell, it is much more attractive to interface external state-of-the-art solvers implemented in lower-level languages. That is why we have recently provided an interface to the Gecode FD solver in C++ [12]. In this work we expand considerably upon this initial interface and show how the same external solver can be interfaced in different modes.

<sup>2</sup> The instance requires `s` to belong to the `FDSolver` class, which requires a type `t` to belong to class `Term s`, which requires `s` to belong to class `Solver` itself.

### 3.1 On-line and Off-line Modes

Firstly, we distinguish between *on-line* and *off-line* use. The former means that the constraint model is processed by the MCP framework, in collaboration with the solver, to produce solutions. This mode is used for the original Haskell-based FD solver. The latter concerns *staged compilation*: in the first stage, the FD model is processed by the MCP framework that produces code for the second stage in the solver’s programming language; the stage-2 code produces solutions. This mode was used in the original Gecode backend of [12]. The off-line mode comes with a compilation function  $\langle\!\langle\cdot\rangle\!\rangle :: \text{Model OfflineSolver } a \rightarrow \text{C++}$  instead of the usual `run` function for solvers.

The off-line mode has a clear appeal for performance reasons: it avoids the interpretative overhead when solving the constraint model in the second stage. Of course, there is the compilation overhead of the first stage. We come back to this issue in the next section, where we considerably improve the usefulness of the off-line mode.

The on-line mode is very convenient for programming the search: all the high-level search features of the MCP framework are available. In contrast, our off-line Gecode solver provides a fixed search strategy. A considerable disadvantage of the on-line mode is the interpretative overhead of Haskell, which is confounded by the fact that the FD solver is implemented in Haskell itself.

**New on-line Gecode solver** In this paper we present a new on-line mode for the external Gecode solver. This combines the performance of Gecode with the high-level search features of the MCP framework. The solver type is defined as:

```
newtype OnlineSolver a
  = OnlineSolver { runOnline :: StateT GecodeState IO a }
```

The `OnlineSolver` is a monad composition of:

**the IO monad:** to access the Gecode library through the Haskell Foreign Function Interface (FFI), and

**the StateT GecodeState monad transformer:** to maintain the solver state:

```
data GecodeState = GecodeState { space :: Space
                                , cexpr :: Map FDExpr IntTerm }
```

which consists of a reference to the current Gecode space, and a map to translate FD expressions in the constraint model to constraint variables in the Gecode solver.

The `OnlineSolver` is recognized as an actual solver by the framework with the following instance:

```
instance Solver OnlineSolver where
  type Constraint OnlineSolver = GecodeConstraint
  add c = addOnlineGecode c
  run m = unsafePerformIO $ do state <- newState
```

```

                                evalStateT state (runOnline m)
type Label OnlineSolver      = GecodeState
mark    = get
goto s = copyState s >>= put

```

The supported constraints of this solver are of type `GecodeConstraint`. The `addOnlineGecode` function adds a constraint to the current Gecode space, through the FFI. This involves constructing the constraint arguments, the FD expressions, in the Gecode solver. The `cexpr` helps out here, capturing earlier mappings of constraint variables and other FD expressions already have a representation in the Gecode solver. This results in dynamic common subexpression elimination. Running the solver means running the underlying IO monad and the state transformer, with appropriate initial state.

Finally, for disjunctive models and branches in the search tree, we use the copying technique in Gecode. Thus for the label of a solver state, we simply use the solver state, i.e. the Gecode space, itself. Whenever creating a branch starting from a given space, we install a copy of that space as the current space so as not to affect other branches.

Thanks to this relatively simple instance, we can now use the MCP infrastructure (e.g., a search queue, compositional search transformers and enumeration) for the on-line Gecode solver.

### 3.2 Programmed versus Fixed Search Modes

The new on-line Gecode backend of the framework offloads constraint propagation on the Gecode solver, but still allows the programmer to program and specify the search heuristics through the high-level interface. We call this approach the *programmed search mode*. It has clear advantages in terms of expressivity, but it does incur an interpretative penalty for search, which for many constraint problems has a considerable impact on the overall solving time.

In order to avoid the interpretative overhead for search, we provide a second mode of on-line use, the *fixed search mode*. Just like the off-line Gecode solver, this mode provides a fixed search strategy implemented in C++ for the on-line Gecode solver. In this mode, labelling the model does not produce a whole subtree that is affected by the framework's search heuristics. Instead, a single node is generated on the MCP side that corresponds to many nodes in the Gecode solver which are processed by a fixed search strategy.

## 4 Parameterized Models

Many FD models are naturally parameterized in a problem size and/or other instance-specific integer values. For instance, the n-queens problem is parameterized in the board size, the Golomb ruler problem is parameterized in the ruler size, ...

Such parameterization does not pose any problem for the on-line solvers. The parameterized model is simply written as a *model function* from one (or more)

integer value to an FD model. An `FDModel` is simply a `Model` for an `FDSolver`, that returns a list of solutions.

```
pmodel :: Int -> FDModel s
pmodel = \n -> ...
```

In order to solve the model, the model function is applied to the appropriate values, and the resulting model is handed to the on-line solver. No surprises.

For off-line solvers, we could follow the same technique. However, then we would obtain a non-parameterized off-line executable. Each time we would like to change the parameters, we would have to generate a new off-line executable! That is very costly in terms of compilation times, compared to the on-line solvers. The latter require only one invocation of the Haskell compiler for a parameterized model, while the former requires one invocation of the Haskell compiler and subsequently, for each instantiation of the parameters, an invocation of the C++ compiler. Moreover, the size of the off-line code is dependent on the problem size, because the framework fully flattens the model before generating code. Hence, the larger the problem size, the bigger the generated C++ code, and the longer the C++ compilation times. In summary, a new approach is necessary to make parameterized models practical for off-line solving.

The remainder of this section shows our approach for representing and compiling parameterized models. It has the two desirable properties: 1) a parameterized model requires only a single invocation of the C++ compiler, and 2) the generated code does not depend on the parameter value.

#### 4.1 Parameters

We still represent parameterized models by model functions, but the functions take FD expression terms rather than integers as arguments.

```
pmodel :: FDEExpr s -> FDModel s
pmodel = \n -> ...
```

For brevity, we will omit the type parameters `s` in further signatures mentioning FD expressions and models.

We still retain the above functionality for off-line solvers, as integer values can be lifted to FD expressions using the `Const :: Integer -> FDEExpr` constructor. Moreover, `FDEExpr` is also an instance of the `Num` type class, so integer literals can be supplied directly as arguments: `pmodel 1425`.

Of more interest is of course the treatment of model functions for off-line solving. A model function is compiled by applying it to special `FDEExpr` values that represent *deferred values*. These deferred values will not be known until the C++ stage. We denote a deferred value in the first stage as ``p`, where `p` is the corresponding representation, a C++ `int` variable, in the second stage.

So using these deferred variables, we again obtain an `FDModel` that can be compiled much as before. Only the deferred values require special care. They are mapped to `int` instance variables of the generated C++ class that represents the Gecode constraint model. A new instance of the problem is created

by instantiating an object of that class with the desired integer values for the parameters.

## 4.2 Indexed Constraint Variable Collections

Unfortunately, this is not the end of the story. Parameters of type `FDEExpr` have fewer uses than values of type `Integer`. Indeed, the former can be used as arguments to constraints, but the latter can appear in many useful Haskell library functions as well as several functions of the MCP framework. Perhaps the most essential such function is `exist :: Integer -> ([Term] -> FModel) -> FModel`, which creates the specified number of constraint variables. In many parametric models, the number of constraint variables depends on the parameter value.

However, for the off-line solver, the integer value of the parameter is not available. Thus the actual creation of the list must be deferred from the on-line Haskell phase to the off-line phase. Moreover, we may wish to use a different data structure than a linked list in the off-line phase, such as an array in C++.

Hence, to allow writing models that can be used with both on-line and off-line solvers, we introduce a type `FDCollection s` indexed by the solver type `s`. For on-line solvers like `OvertonFD`, it is defined as an on-line Haskell list:

```
type instance FDCollection OvertonFD = [FDTerm OvertonFD]
```

For off-line solvers like `OfflineGecode`, a deferred collection ``c` is used that only records an identifier `c` of the particular collection:

```
type instance FDCollection OfflineGecode = OfflineCollection Int
```

However, when writing a constraint model that is polymorphic in the solver type, `FDCollection s` acts as an abstract data type that only allows a limited number of operations, supported by both on-line and off-line solvers. The foremost of these operations are:

- `fdexist :: FDEExpr -> (FDCollection -> FModel) -> FModel` creates a new collection of specified size, and acts as a generalization of `exist`. This function is implemented in terms of `exist` for on-line solvers, but creates a new deferred collection for off-line solvers. Note that the size of generated code for the latter is constant (a single array declaration) as opposed to linear like `exist`.
- `(!) :: FDCollection -> FDEExpr -> FDEExpr` returns an element at a given index in the collection. For on-line solvers it is implemented in terms of list indexing `(!)`, but for off-line solvers a term denoting deferred indexing is returned. Then we have that  $\langle\langle c ! i \rangle\rangle = c[\langle\langle i \rangle\rangle]$ .
- `collect :: [FDEExpr] -> FDCollection` turns a list of variables into a collection.

Global constraints form another class of functions that involve collections. These have been modified to support collections instead of Haskell lists:

- `allDiff :: FDCollection -> FDMoDel` all variables in the given collection are mutually distinct.
- `sorted :: FDCollection -> FDMoDel` the given collection is sorted.
- `allin :: FDCollection -> (FDEExpr, FDEExpr) -> FDMoDel` all variables in the given collection have a value between the given lower and upper bounds.

### 4.3 Iteration

Often the above operations for collections are not expressive enough. Instead of imposing global constraints on a collection or indexing specific entries, many models process all elements of a collection one at a time. For this purpose an iteration construct is necessary.

**Iteration Primitives** We introduce in our framework the iteration primitive `foreach :: (FDEExpr, FDEExpr) -> (FDEExpr -> FDMoDel) -> FDMoDel`, whose denotation is:

$$\llbracket \text{foreach } (l, u) f \rrbracket \equiv \bigwedge_{i=l}^u \llbracket f \ i \rrbracket$$

For instance, we write  $\bigwedge_{i=1}^n (c_i > i)$  as:

```
foreach (1,n) $ \i -> (c ! i) @> i
```

For on-line solvers, `foreach` is expanded literally according to its semantics:

```
foreach (l,u) f = conj [ f i | i <- [l..u] ]
```

However, for off-line solvers, we may not know the range of the loop, if it depends on a model parameter. Even if we do know the range, we may choose not to flatten the loop if the range is too large. In these cases, `foreach` is compiled to a C++ `for`-loop:

```
⟨foreach (l,u) f⟩ = for (int i = ⟨l⟩; i =< ⟨u⟩; i++) { ⟨f `i⟩ }
```

So the size of the generated code does not depend on the size of the iteration range.

Because iteration over the whole range, rather than a subrange, of a collection occurs quite frequently, we introduce a second iteration construct `forall :: FDCollection -> (FDTerm -> FDMoDel) -> FDMoDel`, whose denotation is:

$$\llbracket \text{forall } c f \rrbracket \equiv \bigwedge_{v \in c} \llbracket f \ c \rrbracket$$

For instance, we write  $\bigwedge_{v \in c} (v > i)$  as:

```
forall c $ \v -> v @> i
```

Again the on-line solvers' implementation of `forall` is a direct transliteration of the denotation:

```
forall c f = conj [f v | v <- c]
```

For the off-line solver, we can define `forall` in terms of `foreach`, if we provide access to a collection's size with `size :: FDCollection -> FDEExpr`:

```
forall c f = foreach (1, size c) $ \i -> f (c ! i)
```

which means that we get the following C++ code:

```
⟨forall `c f⟩ = for (int i = 0; i < ⟨size `c⟩; i++) { ⟨f `c[i]⟩ }
```

**Example** The following program is a parameterized model of the n-queens problem:

```
nqueens n =                                -- define model function 'nqueens'
  fdexist n $ \q -> do                      -- new collection 'q' of size n
    q 'allin' (1,n)                          -- all variables in range [1..n]
    foreach (1,n) $ \i ->                   -- for i in [1..n]
      foreach (1,i) $ \j -> do              -- for j in [1..i]
        q!i @/= q!j                          -- \
        q!i + i @/= q!j + j                  -- | constraints
        q!i - i @/= q!j - j                  -- /
    return q                                  -- return result collection
```

Except for import statements and a main function that inputs the parameter value, calls the solver and outputs results, this is a fully working Haskell program.

**Derived Iteration Constructs** Finally, collections need a range of utility functions like those supported on Haskell lists:

- `fdmap :: (FDEExpr -> FDEExpr) -> FDCollection -> Model s FDCollection` transforms each element of a collection using a specified function, similar to the standard Haskell function `map`.
- `fdfold :: (FDEExpr -> FDEExpr -> Model s FDEExpr) -> FDCollection -> Model s FDEExpr` folds a collection to a single expression, similar to the standard Haskell function `foldl`.
- `fdappend :: FDCollection -> FDCollection -> Model s FDCollection` concatenates two collections, similar to the standard Haskell operator `(++)`.

Currently, we implement such utility functions on top of `fdexist` and `foreach`. For example, `fdmap` is implemented as follows:

```
fdmap f c =
  fdexist (size c) $ \result -> do
    foreach (1,size c) $ \i ->
      result!i @= f (c!i)
    return result
```

While this generic implementation is easy to define and works for both on-line and off-line solvers, it does introduce (`size c`) superfluous constraint variables. We will see in the evaluation section that this leads to performance degradation. Hence, we will add these functions as additional primitives in the framework, so solvers can provide their own optimized implementations.

## 5 Evaluation

In order to evaluate the two new extensions, the on-line Gecode solver support and the support for parameterized models, existing benchmarks for Gecode have been ported to FD-MCP. Tables 1 and 2 show the results. Lines of code (LoC) are measured using SLOccount<sup>3</sup>, the timings in seconds are average CPU times over multiple runs.<sup>4</sup>

### 5.1 Solving Results

The first table lists the absolute timings for the original C++ benchmark, and the runtimes of the MCP versions relative to original benchmark. The columns show respectively: 1) the name of the benchmark and the parameter value (if any), 2) the runtime (in seconds) for the Gecode benchmark in C++, and the relative runtimes of 3) the C++ code generated in off-line mode, 4) the parametrized C++ code in off-line mode, 5) the on-line Gecode solver in programmed search mode, 6) the on-line Gecode solver in fixed search mode, and 7) the on-line Haskell-only solver. All of columns 3)-7) are based on the same MCP model. The `-` entries denote out of space, while `+` entries denote a time out (no result after 5 minutes).

Any relative timings close to 100% indicate that the corresponding MCP mode is a valid alternative for direct Gecode implementation in terms of efficiency. Clearly, the pure Haskell solver cannot compete with a native Gecode implementation. Hence, it has been worthwhile to invest in Gecode backends for MCP.

A few times we observe that the compiled code generated in MCP off-line mode is slightly faster than the original Gecode benchmark. This is likely due to start-up overhead where the absolute runtime is only a few milliseconds.

Finally, the MCP versions of the *partition* and *magicsquare* benchmarks depend heavily on `fdmap`, `fdfold` and `fdappend`, which introduce auxiliary constraint variables responsible for the dramatic runtime increase. *magicsquare* introduces  $4n^2 + 4n$  superfluous variables, *partition*  $8n$ . This suggests that optimized versions for these functions are essential to be competitive.

<sup>3</sup> <http://www.dwheeler.com/sloccount/>

<sup>4</sup> Benchmarks have been performed on a 64-bit Ubuntu 9.04 system using a 1.67GHz Intel® Core™2 Duo T5500 processor, with 1GiB RAM. Software versions: GHC 6.10.3, GNU G++ 4.3.3, Gecode 3.1.0.

Benchmark	Gecode (s)	MCP (%)					
		Off-line			On-line		
		C++	C++	param C++	Gecode	Gecode	pure Haskell
				-	+srch.		
allinterval	4	0.006s	85.4%	85.4%	113%	112%	122%
	8	0.006s	91.5%	92.0%	223%	125%	964%
	13	3.52s	108%	108%	462%	94.0%	5640%
	15	120s	107%	108%	-	94.7%	+
queens	5	0.006s	83.6%	84.0%	131%	127%	180%
	13	0.007s	89.3%	89.6%	249%	222%	1870%
	27	0.008s	80.0%	80.7%	65000%	806%	+
	100	0.057s	45.2%	44.9%	+	2400%	+
partition	4	0.006s	87.1%	87.3%	181%	169%	212%
	8	0.007s	118%	118%	294%	216%	995%
	16	0.047s	6500%	6500%	31000%	8200%	210000%
	20	0.15s	49000%	48000%	-	140000%	-
magicsquare	3	0.006s	87.8%	88.1%	211%	203%	267%
	4	0.019s	34.2%	34.2%	104%	91.6%	288%
	5	0.83s	8.9%	9.0%	64.5%	7.3%	1300%
	6	0.007s	64000%	64000%	-	35000%	+

**Table 1.** Timings

## 5.2 Compilation Results

The columns of Table 2 show the name and parameter value of benchmarks, and the number of lines of code and their compilation times for the original C++ benchmark, the benchmark implemented in Haskell using MCP, and the generated C++ code both with and without parameters (unfolded).

These results clearly support two conclusions:

1. Models written in MCP are more concise than in Gecode, and
2. Parametrized generated code avoids parameter-dependant code sizes.

## 6 Related and Future Work

There is wide range of CP systems and languages. For lack of space we only mention a few. We classify them according to the distinction made in Section 1. A more extensive overview of related work can be found in [10].

*Stand-alone modeling languages* Zinc [8] is a stand-alone modeling language. Model transformations and compilation processes to different constraint solver backends are implemented in a second language, Cadmium, which is based on ACD term rewriting [3].

Rules2CP [4] is another stand-alone modeling language. The compilation of Rules2CP to SICStus Prolog is also specified by rewrite rules.

Benchmark	Lines of code					Compilation time(s)			
	C++	MCP			C++	MCP			
		C++	unfold.	C++		C++	unfold.	C++	
allinterval	3	52	19	61	71	3.3	0.22	2.7	2.0
	15				179				2.7
queens	4	80	14	53	79	3.4	0.21	2.5	2.1
	100				534				>20
partition	4	74	34	103	103	3.3	0.21	3.1	2.2
	20				295				4.5
magicsquare	3	62	35	94	92	3.4	0.22	2.9	2.8
	6				206				3.1

**Table 2.** Lines of code and compilation times

*Constraint Programming API's* Two other functional programming languages which provide CP support are Alice ML<sup>5</sup> and FaCiLe [1]. While Alice ML provides a run-time interface to Gecode [6], FaCiLe uses its own constraint solver in OCaml. Both provide a rather low-level and imperative API, which corresponds to the C++ API of Gecode in the case of Alice ML, and relies on side effects. Neither supports alternative backends or model transformations.

*Integrations* Cipriano et al. [2] translate constraint models written in both Prolog CLP(FD) and (Mini)Zinc to Gecode via an intermediate language called CNT without loop constructs. The transformation from CNT to Gecode is implemented in Haskell. In order to avoid the Gecode code blow-up, it attempts to identify loops in the unrolled CNT model. It also performs a number of simplifications in the model. Our approach is much more convenient and efficient, providing explicit looping constructs and compiling these directly without intermediate loop unrolling, and with strong guarantees that loops remain loops.

*Future work* The benchmarks clearly indicate that additional iteration primitives must be added to the framework, in order to be competitive with Gecode. The support for collections should also be further extended to *multi-dimensional indexing*, which is quite convenient for modelling grid-based problems like sudoku, and *collection parameters* for providing a variable number of deferred data such as supply and demand quantities in a transportation problem.

## 7 Conclusions

We have shown how to link the FD-MCP framework with Gecode to allow efficient on-line solving of constraint problems modeled using it. Furthermore, we added deferred parameters and indexable collections to the provided abstractions, allowing shorter and more useful off-line code to be generated. These

<sup>5</sup> <http://www.ps.uni-sb.de/alice>

extensions were implemented<sup>6</sup> and benchmarks show that there is often only a small performance penalty compared to native C++ implementations.

**Acknowledgments** We are grateful to Peter Stuckey for his helpful comments.

## References

1. N. Barnier. *Application de la programmation par contraintes à des problèmes de gestion du trafic aérien*. PhD thesis, Institut National Polytechnique de Toulouse, December 2002. <http://www.recherche.enac.fr/opti/papers/thesis/>.
2. R. Cipriano, A. Dovier, and J. Mauro. Compiling and executing declarative modeling languages to Gecode. In M. G. de la Banda and E. Pontelli, editors, *ICLP*, volume 5366 of *LNCS*, pages 744–748, 2008.
3. G. J. Duck, P. J. Stuckey, and S. Brand. ACD term rewriting. In S. Etalle and M. Truszczynski, editors, *ICLP*, volume 4079 of *LNCS*, pages 117–131, 2006.
4. F. Fages and J. Martin. From Rules to Constraint Programs with the Rules2CP Modelling Language. In *Recent Advances in Constraints*, LNAI, 2009.
5. A. J. Fernandez, T. Hortala-Gonzalez, F. Saenz-Perez, and R. Del Vado-Virseda. Constraint functional logic programming over finite domains. *Theory Pract. Log. Program.*, 7(5):537–582, 2007.
6. Gecode Team. Gecode: Generic constraint development environment, 2006. Available from <http://www.gecode.org>.
7. M. Hanus (ed.). Curry: An integrated functional logic language (vers. 0.8.2). Available at <http://www.curry-language.org>, 2006.
8. K. Marriott et al. The design of the Zinc modelling language. *Constraints*, 13(3):229–267, 2008.
9. S. Peyton Jones et al. The Haskell 98 language and libraries: The revised report. *Journal of Functional Programming*, 13(1):0–255, Jan 2003.
10. T. Schrijvers, P. Stuckey, and P. Wadler. Monadic Constraint Programming. *J. Func. Prog.*, 19(6):663–697, 2009.
11. P. Wadler. Monads for functional programming. In *Advanced Functional Programming*, pages 24–52, London, UK, 1995.
12. P. Wuille and T. Schrijvers. Monadic Constraint Programming with Gecode. In *Proceedings of the 8th International Workshop on Constraint Modelling and Reformulation*, pages 171–185, 2009.

---

<sup>6</sup> Available at <http://www.cs.kuleuven.be/~toms/MCP/>